1　Mapleの数式処理

数式の変形に関するコマンドをまとめています．手で直すほうが圧倒的に早くうまい場合が多いです．しかし，テイラー展開や，複雑な積分公式，三角関数とexp関数の変換などの手間のかかるところを，Mapleは間違いなく変形してくれます．ここで示したコマンドを全て覚える必要は全くありません．というか忘れるものです．ここでは，できるだけコンパクトにまとめて，悩んだときに参照できるようにしたつもりです．

1.1　コマンド解説

1.1.1　コマンド表

まず数式処理でよく使うコマンドをいくつかの範疇に分類してまとめておきます．このほかにも前節までに示した，solve(解), diff(微分), int(積分), は頻繁に数式の導出・変形に登場します．

表 1: 式の操作に関係するコマンド

式の変形	式の分割抽出	代入，置換，仮定
simplify:簡単化	lhs, rhs:左辺，右辺	subs:一時的代入
expand:展開	numer, denom:分子，分母	assume:仮定
factor:因数分解	coeff:係数	assuming:一時的仮定
normal:約分・通分	nops, op	assign:値の確定
combine:公式でまとめる		about:仮定の中身
collect:次数でまとめる		anames('user'):使用変数名
sort:昇べき，降べき		restart,a:='a':初期化
convert:形式の変換		

また，rem(商), quo(余り), limit(極限) なども使います．

1.1.2　コマンド使用例

式の変形

expand: 展開 `expand(`*exp1*`)`

factor: 因数分解 `factor(`*exp1*`)`

normal: 約分・通分 `normal(`*exp1*`)`

combine: 公式でまとめる `combine(`*exp1*`)`

collect: 次数でまとめる `collect(`*exp1*`,x)`

convert: 形式の変換 `convert(`*exp1*`,`*opt*`)`

```
> convert(sin(x),exp);
> convert(sinh(x),exp);
> convert(exp(I*x),trig);
> convert(1/(x-1)/(x+3),parfrac);
```

$$-1/2\,i\left(e^{ix}-\left(e^{ix}\right)^{-1}\right)$$
$$1/2\,e^{x}-1/2\,\left(e^{x}\right)^{-1}$$
$$\cos\left(x\right)+i\sin\left(x\right)$$
$$1/4\,\left(x-1\right)^{-1}-1/4\,\left(x+3\right)^{-1}$$

表 2: convert による形式の変換

opt	意味
polynom	級数を多項式に変換
trig	三角関数に変換
exp	指数関数形式に変換
parfrac	部分分数に変換
rational	浮動小数点数を有利形式に変換

simplify:簡単化 `simplify(`*exp1*`)`, `simplify(`*exp1*`,` 副関係式`)`

```
> exp1:=3*sin(x)^3-sin(x)*cos(x)^2;
```

$$exp1\ :=\ 3\,\left(\sin\left(x\right)\right)^{3}-\sin\left(x\right)\left(\cos\left(x\right)\right)^{2}$$

```
> simplify(exp1);
```

$$-\left(4\,\left(\cos\left(x\right)\right)^{2}-3\right)\sin\left(x\right)$$

```
> simplify(exp1,{cos(x)^2=1-sin(x)^2});
```

$$-\sin\left(x\right)+4\,\left(\sin\left(x\right)\right)^{3}$$

sort:昇べき，降べきソート `sort(`*exp1*`)`, `sort(`*exp1*`,[x,y])`, `sort(`*exp1*`, [x],opts);`*opts=tdeg,plex,ascending,or descending*（それぞれ総次数順序，辞書式順序，昇順，降順）

```
> exp1:=x^3+4*x-3*x^2+1:
> sort(exp1);
> sort(exp1,[x],ascending);
```

$$x^{3}-3\,x^{2}+4\,x+1$$
$$1+4\,x-3\,x^{2}+x^{3}$$

```
> exp2:=x^3-3*x*y+4*x^2+y^2:
> sort(exp2);
> sort(exp2,[x]);
> sort(exp2,[y],descending);
```

$$x^3+4x^2-3xy+y^2$$
$$x^3+4x^2-3yx+y^2$$
$$y^2-3xy+x^3+4x^2$$

式の分割抽出

lhs, rhs: 左辺，右辺 `lhs(`*exp1*`=`*exp2*`)`

numer, denom: 分子，分母 `numer(`*exp1*`/`*exp1*`)`

coeff: 係数 `coeff(exp1,x^2)`

op,nops: 要素の取り出し, 要素数 `op(`*exp1*`)`, `nops(`*exp1*`)`

代入，置換，仮定

subs: 一時的代入 `subs(関係式,`*exp1*`)`

```
> exp1:=x^2-4*x+4;
> subs(x=a+2,exp1);
```

$$exp1 := x^2-4\,x+4$$
$$(a+2)^2-4\,a-4$$

assume: 仮定 assume(関係式)

assuming: 一時的仮定 exp1 assuming 関係式

```
> exp1:=x^2-4*x+4;
> sqrt(exp1);
> sqrt(exp1) assuming x>2;
```

$$exp1 := x^2-4\,x+4$$
$$\sqrt{(-2+x)^2}$$
$$-2+x$$

additionally: assume に加えての仮定

assign: solve で解いた値の確定.

about: assume で仮定した内容の確認

restart,a='a': 値の初期化

anames('user'): 使っているユーザー定義変数の確認

級数展開

series: 級数展開，series(*exp1*,x,4)

```
> series(exp(x),x);
```

$$1 + x + 1/2\,x^2 + 1/6\,x^3 + 1/24\,x^4 + \frac{1}{120}\,x^5 + O\left(x^6\right)$$

```
> series(sin(x),x=Pi/3,2);
```

$$1/2\,\sqrt{3} + 1/2\,x - 1/6\,\pi + O\left(\left(x - 1/3\,\pi\right)^2\right)$$

```
> convert(%,polynom);
```

$$1/2\,\sqrt{3} + 1/2\,x - 1/6\,\pi$$

省略操作

||: 連結作用素，前後の変数をくっつけて新たな変数とする．

```
> a||1;
> a||b;
```

$$\mathit{a1}$$

$$ab$$

seq: for-loop の単純表記

```
> seq(i,i=0..3);
```

$$0,\ 1,\ 2,\ 3$$

map: 関数の割り当て

これらを組み合わせて実行すると，効率的に式を扱うことが出来る．

```
> map(sin,[seq(a||i,i=0..3)]);
```

$$[\sin\left(\mathit{a0}\right), \sin\left(\mathit{a1}\right), \sin\left(\mathit{a2}\right), \sin\left(\mathit{a3}\right)]$$

その他

add, mul: 単純な和，積

sum, product: 公式にも対応した和，積．

```
> add(x^i,i=1..3);
> add(x^i,i=1..n);
```

$$x + x^2 + x^3$$

```
Error, unable to execute add
```

```
> mul(x^i,i=1..3);
mul(x^i,i=1..n);
```

$$x^6$$

```
Error, unable to execute add
```

```
> sum(x^i,i=1..3);
sum(x^i,i=1..n);
```

$$x + x^2 + x^3$$
$$\frac{x^{n+1}}{x-1} - \frac{x}{x-1}$$

```
> product(x^i,i=1..3);
> product(x^i,i=1..n);
```

$$x^6$$
$$\prod_{i=1}^{n} x^i$$

limit: 極限

```
> limit(exp(-x),x=infinity);
> limit(tan(x),x=Pi/2,left);
> limit(tan(x),x=Pi/2,complex);
```

$$0$$
$$\infty$$
$$-\infty + \infty * I$$

1.2 鉄則とその具体例

Maple をはじめとする数式処理ソフトの習得にあたって初心者がおちいる共通の過ちを回避する鉄則があります．それは

鉄則 0 restart をかける： 続けて入力すると前の入力が生きている．違う問題へ移るときやもう一度入力をし直すときには restart を入力して初期状態からはじめる．

鉄則 1 出力してみる： 多くのテキストではページ数の関係で出力を抑止しているが，初心者が問題を解いていく段階ではデータやグラフをできるだけ多く出力する．

鉄則 2 関数に値を代入してみる： 数値が返ってくるべき時に変数があればどこかで入力をミスっている．plot で Plotting error,empty plot が出た場合にチェック．

鉄則 3 内側から順に入力する： 長い入力は内側の関数から順に何をしているか確認しながら打ち込む．

です．

例えば，複雑な積分として以下のような問題があったとします．

$$\int_{-\infty}^{\infty} x e^{-\beta c x^2} \left(1 + \beta g x^3\right) dx \tag{1}$$

最新の Maple では改良が施されていて，このような複雑な積分も一発で

```
> f1:=unapply(x*exp(-beta*c*x^2)*(1+beta*g*x^3),x);
```

$$f1 := x \mapsto xe^{-\beta\, cx^2}\,(1+\beta\, gx^3)$$

```
> int(f1(x),x=-infinity..infinity);
```

$$PIECEWISE\left([3/4\,\frac{g\sqrt{\pi}}{\beta\, c^2\sqrt{\beta\, c}}, csgn\,(\beta c)=1], [\infty, otherwise]\right)$$

と求まるようになっています．ここではβc が正の場合 (csgn(βc)=1) とそれ以外の場合 (otherwise) に分けて答えを返しています．しかしこのような意図したきれいな結果を Maple が返してくれるとは限りません．これだけですと，なにかうまくいかないときにお手上げになってしまいます．このようなきれいで簡単な結果に行き着く前の，裏でおこなういくつかの予備計算を省略せずに示します．先ず鉄則 0 にしたがって restart をかけ，関数を定義します．

```
> restart;
> f1:=unapply(x*exp(-beta*c*x^2)*(1+beta*g*x^3),x);
```

$$f1 := x \mapsto xe^{-\beta\, c\, x^2}\,(1+\beta\, gx^3)$$

次には鉄則 1 にしたがって積分する前にどのような関数かプロットしてみます．そのまま

```
> plot(f1(x),x=-10..10);
```

とすると

```
Warning, unable to evaluate the function to numeric values in the region;
see the plotting command's help page to ensure the calling sequence is correct

Error, empty plot
```

と怒られます．これは鉄則 2 にあるとおり，数値を代入すれば

```
> f1(10);
```

$$10\, e^{-100\,\beta\, c}\,(1+1000\,\beta\, g)$$

で，beta,c,g などのパラメータの値が入っていないためとわかります．適当に

```
> c:=1; g:=0.01; beta:=0.1;
```

$$c := 1$$
$$g := 0.01$$
$$\beta := 0.1$$

として，変数に値を代入し，再度プロットを試みると

```
> plot(f1(x),x=-10..10);
```

とグラフを書いてくれます．これから $-\infty..\infty$ の積分がなんらかの値を取ることが期待できます．さらに鉄則 3 にしたがって，式を頭から打ち込むのではな



く内側からみていきます．これは問題を解いていく時に思考が必ずたどるであろう順番に相当します．先ず変数に入れた数値をクリアします．

```
>  c:='c'; g:='g'; beta:='beta';
```

$$c := c$$
$$g := g$$
$$\beta := \beta$$

不定積分でこの関数が積分できることを確認し，

```
>  int(f1(x),x);
```

$$-1/2\,\frac{1}{e^{\beta\,c x^2}\beta\,c}+\beta\,g\left(-1/2\,\frac{x^3e^{-\beta\,c x^2}}{\beta\,c}+3/2\left(-1/2\,\frac{xe^{-\beta\,c x^2}}{\beta\,c}+1/4\,\frac{\sqrt{\pi}\mathit{erf}\left(\sqrt{\beta\,c}x\right)}{\beta\,c\sqrt{\beta\,c}}\right)\beta^{-1}c^{-1}\right)$$

次に x=-alpha..alpha の定積分を実行してみます．

```
>  int(f1(x),x=-alpha..alpha);
```

$$-1/4\,\frac{g\left(4\,\alpha^3e^{-\beta\,c\alpha^2}\beta\,c\sqrt{\beta\,c}+6\,\alpha\,e^{-\beta\,c\alpha^2}\sqrt{\beta\,c}-3\,\sqrt{\pi}\mathit{erf}\left(\sqrt{\beta\,c}\alpha\right)\right)}{\beta\,c^2\sqrt{\beta\,c}}$$

さらに alpha$\rightarrow\infty$ としてみます

```
>  limit(int(f1(x),x=-alpha..alpha),alpha=infinity);
```

$$\lim_{\alpha\to\infty}-1/4\,\frac{g\left(4\,\alpha^3e^{-\beta\,c\alpha^2}\beta\,c\sqrt{\beta\,c}+6\,\alpha\,e^{-\beta\,c\alpha^2}\sqrt{\beta\,c}-3\,\sqrt{\pi}\mathit{erf}\left(\sqrt{\beta\,c}\alpha\right)\right)}{\beta\,c^2\sqrt{\beta\,c}}$$

ところがこれでは答えを返してくれません．積分した後のそれぞれの項を見ると beta*c>0 を仮定すれば簡単になることが分かります．このような変数の仮定 (assume) は

```
>  assume(beta*c>0);
```

でおこないます．結果として最初に出した解答

```
>  limit(int(f1(x),x=-alpha..alpha),alpha=infinity);
```

$$3/4\,\frac{\sqrt{\pi}g}{\beta\,c^2\sqrt{\beta\,c}}$$

が導かれるのです．数式処理ソフトでの数式処理とは数式処理ソフトが『自動的にやって』くれるのではなく，実際に紙と鉛筆で解いていく手順を数式処理ソフトに『やらせる』のだということを肝に銘じてください．

1.3 実戦例

どうしても解かなければならない課題を前にコマンドリファレンスのあちこちを参照しながら解いていくのが数式処理を修得する最速法です．しかし，なかなか共通する適当な課題がありません．以下では大学物理の初歩として比較的多くの人が出会う「熱膨張係数の導出 (キッテル固体物理)」と「トンネル効果 (シッフ量子力学)」を取り上げます．著者がどのようないじくり方をしているかを眺めてみてください．その前に「式のフォローのデフォルト」です．

1.3.1 式のフォローのデフォルト

Maple で実際に数式をいじる状況というのは，ほとんどの場合が既知の数式変形のフォローです．例えば，論文で「(1) 式から (2) 式への変形は自明である」とかいう文章で済ましている変形が本当にあっているのかを確かめたい時です．一番単純なやり方は自明と言われた前後の式が一致していることを確かめるだけで十分です．最も単純には

```
> ex1:=(x-3)^4;
```

$$ex1 := (x-3)^4$$

```
> ex2:=x^4-12*x^3+54*x^2-108*x+81;
```

$$ex2 := x^4 - 12\,x^3 + 54\,x^2 - 108\,x + 81$$

```
> expand(ex1-ex2);
```

$$0$$

というように，変形の前後の式を手入力してその差を expand した結果が 0 か否かでします．0 ならば式の変形は保証されていますから，その導出が間違いでなく誤植などもないことだけは信じられます．ただ，これだけでは変形の哲学や技法が身に付くわけではありませんので，あくまでも苦し紛れのデフォルトであることは心に留めておいてください．リバースエンジニアリングがいかに楽かが分かります．

1.3.2 熱膨張係数の導出

（参考 キッテル著 固体物理学入門 宇 野良清他訳，丸善 1978 ）熱膨張 (thermal expansion) は原子間ポテンシャルの二次以上の項によって現れます．平衡点から

の原子の変位 x のポテンシャルエネルギーを

$$U(x) = cx^2 - gx^3 \tag{2}$$

と取ることができます．二次までの項では古典的な調和振動子を表し，熱膨張は現れません．x^3 の項は原子間相互作用の非対称性を表し，この項が熱膨張係数と直接かかわってきます．

有限温度での平均の変位は，ボルツマン分布関数を計算することで求まります．平均の位置 x は，熱力学的な確率で重みづけられ，

$$\langle x \rangle = \frac{\int_{-\infty}^{\infty} x \exp\left(-\beta U(x)\right) dx}{\int_{-\infty}^{\infty} \exp\left(-\beta U(x)\right) dx} \tag{3}$$

で計算できます．ここで $\beta \equiv 1/(k_{\mathrm{B}}T)$ です．この積分を実行すると

$$\langle x \rangle = \frac{3g}{4\beta c^2} \tag{4}$$

となります．

最後の導出が問題です．ここでは先ず近似によって式を簡単化します．これはお決まりの Taylor 展開です．味噌はポテンシャルエネルギーの項で二次の項はそのまま残し，三次以上を展開することです．実際に Maple で展開してみると，

```
> restart;
> U:=c*x^2-g*x^3;
> eU:=expand(exp(-beta*U));
```

$$U := cx^2 - gx^3$$

$$eU := \frac{e^{\beta\, gx^3}}{e^{\beta\, cx^2}}$$

```
> ex:=convert(series(numer(eU),x,4),polynom);
```

$$ex := 1 + \beta\, gx^3$$

```
> f1:=ex/denom(eU);
```

$$f1 := \frac{1+\beta\, gx^3}{e^{\beta\, cx^2}}$$

という式が導かれます．一見簡単に見えるかもしれませんが，一つ一つを内側から入力してコマンド表を参照し，出力を見ながらどのように変形が進んでいくかを確認してください．

次にこの式の積分です．分子と分母を別々に積分してみます．鉄則の具体例で例示した積分です．

```
> den:=int(f1,x=-infinity..infinity);
```

$$den := PIECEWISE\left([\frac{\sqrt{\pi}}{\sqrt{\beta\, c}}, csgn\left(\beta\, c\right) = 1], [\infty, otherwise] \right)$$

```
> num:=int(x*f1,x=-infinity..infinity);
```

$$num := PIECEWISE\left([3/4\,\frac{g\sqrt{\pi}}{\beta\,c^2\sqrt{\beta\,c}}, csgn\,(\beta c)=1], [\infty, otherwise]\right)$$

$\beta c > 0$ の場合とそれ以外の結果を別々に返しています．題意から明らかなように $\beta c > 0$ ですので，それを仮定 (assume) して，分子÷分母を実行します．

```
> assume(beta*c>0):
> num/den;
```

$$3/4\,\frac{g}{\beta\,c^2}$$

1.3.3 トンネル効果

（参考：シッフ著 量子力学（井上 健訳），吉岡書店 1970．小出昭一郎著 基礎物理学選書 5A 一量子力学，裳華房 1969．量子効果の基礎的な例である 1 次元の



トンネル効果についての式を導いてみましょう．上図のようなポテンシャルを考えます．詳しい解説は成書を参考にしていただくとして，以下で扱う数式の簡単な説明だけをしておきます．まず，ポテンシャルエネルギー $V(x) = 0$ の領域での波動方程式は

$$-\frac{\hbar^2}{2m}\frac{d^2\varphi(x)}{dx^2} = \varepsilon\varphi(x) \tag{5}$$

ですから，波動関数は

$$\begin{aligned} x \le 0 \quad \text{では} \quad \varphi(x) &= A\exp(ikx) + B\exp(-ikx) \\ x \ge a \quad \text{では} \quad \varphi(x) &= C\exp(ikx). \end{aligned} \tag{6}$$

ここで $k = \sqrt{2m\varepsilon/\hbar^2}$ は波数ベクトルの大きさです．

ポテンシャルの壁の内側では $\varepsilon \lesseqgtr V_0$ によって事情が変わってきます．$\varepsilon \ge V_0$ では $\kappa = \sqrt{2m(\varepsilon - V_0)/\hbar^2}$ と定義すると，波動関数は

$$0 \le x \le a \text{ では } \ \varphi(x) = F\exp(i\kappa x) + G\exp(-i\kappa x) \tag{7}$$

となります．波動関数は粒子の座標に関する滑らかな連続関数でなければならないという条件を $x=0$ と $x=a$ に適用すると，条件はそれぞれ

$$
\begin{aligned}
&x=0 \quad \text{で } \varphi(x) \text{ が連続：} && A+B=F+G \\
&x=0 \quad \text{で } \varphi'(x) \text{ が連続：} && k(A-B)=\kappa(F-G) \\
&x=a \quad \text{で } \varphi(x) \text{ が連続：} && F\exp(i\kappa a)+G\exp(-i\kappa a)=C\exp(ika) \\
&x=a \quad \text{で } \varphi'(x) \text{ が連続：} && \kappa F\exp(i\kappa a)-\kappa G\exp(-i\kappa a)=kC\exp(ika)
\end{aligned} \tag{8}
$$

で与えられます．係数が 5 個で，方程式が 4 個ですから，それぞれの係数の比だけが求まります．これらから F, G を消去して，B/A 入射波と反射波の複素振幅の比，および C/A 入射波と透過波の複素振幅の比が求まります．これらの二乗が反射係数と透過係数にほかなりません．結果は

$$
\begin{aligned}
\left|\frac{B}{A}\right|^2 &= \left[1+\frac{4k^2\kappa^2}{(k^2-\kappa^2)^2\sin^2\kappa a}\right]^{-1} = \left[1+\frac{4\varepsilon(\varepsilon-V_0)}{V_0^2\sin^2\kappa a}\right]^{-1} \\
\left|\frac{C}{A}\right|^2 &= \left[1+\frac{(k^2-\kappa^2)^2\sin^2\kappa a}{4k^2\kappa^2}\right]^{-1} = \left[1+\frac{V_0^2\sin^2\kappa a}{4\varepsilon(\varepsilon-V_0)}\right]^{-1}
\end{aligned} \tag{9}
$$

となります．

$0<\varepsilon<V_0$ ならば，$\alpha=\sqrt{2m(V_0-\varepsilon)/\hbar^2}$ として波動関数は

$$
0 \le x \le a \text{ では} \varphi(x)=F\exp(\alpha x)+G\exp(-\alpha x) \tag{10}
$$

です．同様な計算によって

$$
\left|\frac{C}{A}\right|^2 = \left[1+\frac{V_0^2\sinh^2\alpha a}{4\varepsilon(\varepsilon-V_0)}\right]^{-1} \tag{11}
$$

となります．$mV_0a^2/\hbar^2 = 8$ の場合の透過率を求めると下図のようになります．$|E/V_0|<1$ でも透過波の比強度 $|C/A|^2$ が有限の値を取る現象がトンネル効果です．

実際に C/A を導出して，その振る舞いをプロットさせてみましょう．先ずは波動関数の定義です．

```
> restart;
> psi1:=x->A*exp(I*k*x)+B*exp(-I*k*x);
```

$$psi1 := x \mapsto Ae^{ikx}+Be^{-ikx}$$

```
> psi2:=x->E*exp(I*kappa*x)+F*exp(-I*kappa*x);
```

$$psi2 := x \mapsto Ee^{i\kappa x}+Fe^{-i\kappa x}$$

```
> psi3:=x->C*exp(I*k*x);
```

$$psi3 := x \mapsto Ce^{ikx}$$



次に $x = 0, x = a$ での 0 次，1 次微分の連続条件を入れます．

```
> eq1:=psi1(0)=psi2(0);
```

$$eq1 := A + B = E + F$$

```
> eq2:=simplify(subs(x=0,diff(psi1(x),x))=subs(x=0,diff(psi2(x),x)));
```

$$eq2 := iAk - iBk = iE\kappa - iF\kappa$$

```
> eq3:=psi2(a)=psi3(a);
```

$$eq3 := Ee^{i\kappa a} + Fe^{-i\kappa a} = Ce^{ika}$$

```
> eq4:=simplify(subs(x=a,diff(psi2(x),x))=subs(x=a,diff(psi3(x),x)));
```

$$eq4 := i\kappa\left(Ee^{i\kappa a} - Fe^{-i\kappa a}\right) = iCke^{ika}$$

この 4 つの方程式を {A,B,C,E,F} について解きます．

```
> solve(eq1,eq2,eq3,eq4,A,B,C,E,F);
```

$$\begin{aligned}
C &= C, B = -1/4\,\frac{\left(\kappa^2 - \kappa^2\left(e^{i\kappa a}\right)^2 - k^2 + k^2\left(e^{i\kappa a}\right)^2\right)Ce^{ika - i\kappa a}}{\kappa\,k},\\
F &= -1/2\,\frac{C\left(-\kappa + k\right)e^{ika + i\kappa a}}{\kappa},\\
A &= 1/(4\kappa\,k)\,C\left(e^{ika - i\kappa a}\kappa^2 - e^{ika - i\kappa a}\kappa^2\left(e^{i\kappa a}\right)^2 + e^{ika - i\kappa a}k^2 + e^{ika - i\kappa a}k^2\left(e^{i\kappa a}\right)^2\right.\\
&\quad\left.+2\,e^{ika - i\kappa a}k\kappa + 2\,e^{ika + i\kappa a}k\kappa - 2\,e^{ika + i\kappa a}k^2\right),\\
E &= 1/2\,\frac{\left(\kappa + k\right)Ce^{ika - i\kappa a}}{\kappa}
\end{aligned}\tag{12}$$

```
> assign(%);
```

assign で確定しておきます．次に実際に A/C とその複素共役 (conjugate) との積から比強度を出すのですが，うまく複素共役を取れるように係数 kappa, k,a が実数を仮定していることを Maple に教えてやります．

```
> assume(kappa,real);assume(k,real),assume(a,real);
```

conjugateを取ってから三角関数(trig)へ変換すると式がややこしくなりましたので，先にconvertしてから複素共役を取ります．

```
> CC:=convert(A/C,trig):
> CC1:=combine(conjugate(CC)*CC);
> #CC1:=convert(CC,trig);
```

$$CC1 := \frac{1/8\,k^4+1/8\,\kappa^4+3/4\,\kappa^2k^2-1/8\,k^4\cos(2\,\kappa\,a)+1/4\,\kappa^2k^2\cos(2\,\kappa\,a)-1/8\,\kappa^4\cos(2\,\kappa\,a)}{\kappa^2k^2}$$

後はその単純化です．

```
> C_num:=simplify(expand(numer(CC1)),
> cos(kappa*a)^2=1-sin(kappa*a)^2,
> cos(k*a)^2=1-sin(k*a)^2);
```

$$C\_num := 8\,\kappa^2k^2+\left(2\,k^4-4\,\kappa^2k^2+2\,\kappa^4\right)\left(\sin\left(\kappa\,a\right)\right)^2$$

```
> C_den:=denom(CC1);
```

$$C\_den := 8\,\kappa^2k^2$$

```
> saa:=sin(kappa*a);
```

$$saa := \sin\left(\kappa\,a\right)$$

```
> CC2:=collect(C_num/C_den,saa);
```

$$CC2 := 1+1/8\,\frac{\left(2\,k^4-4\,\kappa^2k^2+2\,\kappa^4\right)\left(\sin(\kappa\,a)\right)^2}{\kappa^2k^2}$$

これで透過率が求まりました．次に，いくつかの条件式からk,kappa,aを求めます．

```
> NN:=8;
> a2:='a2';
> a2:=solve(m*V0*a2/h2=NN,a2);
> kappa2:=2*m*(epsilon-V0)/h2;
> k2:=2*m*epsilon/h2;
```

$$NN := 8$$

$$a2 := a2$$

$$a2 := 8\,\frac{h2}{m\,V0}$$

$$kappa2 := 2\,\frac{m(\epsilon-V0)}{h2}$$

$$k2 := 2\,\frac{m\epsilon}{h2}$$

```
> CC3:=simplify(subs(k=sqrt(k2),kappa=sqrt(kappa2),
> coeff(CC2,saa^2)));
```

$$CC3 := -1/4\,\frac{V0^2}{(V0-\epsilon)\epsilon}$$

```
> CC4:=simplify(subs(epsilon=x*V0,CC3));
```

$$CC4 := 1/4\,\frac{1}{(-1+x)x}$$

```
> a2a2:=simplify(subs(epsilon=x*V0,sqrt(kappa2*a2)));
```

$$a2a2 := 4\,\sqrt{-1+x}$$

```
> CC5:=1+CC4*sin(a2a2)^2;
```

$$CC5 := 1 + 1/4\,\frac{\left(\sin\left(4\sqrt{-1+x}\right)\right)^2}{(-1+x)x}$$

この式を関数と定義してその振る舞いをプロットします.

```
> f1:=unapply(CC5,x);
```

$$f1 := x \mapsto 1 + 1/4\,\frac{\left(\sin\left(4\sqrt{-1+x}\right)\right)^2}{(-1+x)x}$$

```
> plot(1/f1(x),x=0..10);
```

この結果が冒頭に示した透過率の図です.

課題

1. **必須課題**
   鉄則の具体例と実戦例 (熱膨張係数の導出とトンネル効果) を (i) コマンドの内側から入力する, (ii) それぞれのコマンドの意味をテキストで参照する, に注意してフォローしなさい.

2. **自由課題**
   実戦例のトンネル効果で, $0 < \varepsilon < V_0$ の場合に得られる `sinh` を含んだ式を導出せよ. 例で示した `sin` を含んだ式の導出を少し換えれば求まる. ただし,

$$\cosh^2(\alpha * a) = 1 + \sinh^2(\alpha * a) \tag{13}$$

   であることに注意.